Hewlett-Packard

HP-19C/HP-29C SOLUTIONS

STATISTICS



INTRODUCTION

This HP-19C/HP-29C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become an expert on your HP calculator.

You will find general information on how to key in and run programs under "A Word about Program Usage" in the Applications book you received with your calculator.

We hope that this Solutions book will be a valuable tool in your work and would appreciate your comments about it.

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

© HEWLETT-PACKARD COMPANY 1977

TABLE OF CONTENTS

ARITHMETIC, GEOMETRIC, HARMONIC AND GENERALIZED MEANS

Arithmetic mean

$$A= \frac{a_1 + \dots + a_n}{n}$$

Geometric mean

$$G= (a_1 a_2 \dots a_n)^{\frac{1}{n}}$$

Harmonic mean

$$H= \frac{n}{\frac{1}{a_1} + \frac{1}{a_2} + \dots + \frac{1}{a_n}}$$

Generalized mean

$$M(t)= \left(\frac{1}{n} \sum_{k=1}^{n} a_k{}^t\right)^{\frac{1}{t}}$$

NOTES: 1. $a_k > 0$, k= 1, 2, ..., n

2. M(1)=A

M (-1)=H

EXAMPLES:

Find A, G, H & M (1) for the set of numbers

{2, 3.4, 3.41, 7, 11, 23}

SOLUTION:

```
        GSB1
 1.00   ST09
 2.00   GSB2
 3.40   GSB2
 3.41   GSB2
 7.00   GSB2
11.00   GSB2
23.00   GSB2
        GSB3
 8.30    ***   A
         R/S
 4.40    ***   H
         R/S
 5.87    ***   G
         R/S
 8.30    ***   M(t)
```

BASIC STATISTICS (TWO VARIABLES)

This program calculates means, standard deviations, covariance and correlation coefficient derived from a set of data points

$$\{(x_i, y_i), i=1, 2, \ldots, n\}$$

means $\bar{x} = \frac{1}{n}\sum_{i=1}^{n} x_i \quad \bar{y} = \frac{1}{n}\sum_{i=1}^{n} y_i$

standard deviations

$$s_x = \sqrt{\frac{\Sigma x_i^2 - n\bar{x}^2}{n-1}}$$

$$\left(\text{or } \sigma_x = \sqrt{\frac{\Sigma x_i^2 - n\bar{x}^2}{n}} = s_x\sqrt{\frac{n-1}{n}}\right)$$

$$s_y = \sqrt{\frac{\Sigma y_i^2 - n\bar{y}^2}{n-1}}$$

$$\left(\text{or } \sigma_y = \sqrt{\frac{\Sigma y_i^2 - n\bar{y}^2}{n}} = s_y\sqrt{\frac{n-1}{n}}\right)$$

covariance

$$s_{xy} = \frac{1}{n-1}\left(\Sigma x_i y_i - \frac{1}{n}\Sigma x_i \Sigma y_i\right)$$

$$\left(\text{or } s_{xy}' = \frac{1}{n}\left[\Sigma x_i y_i - \frac{1}{n}\Sigma x_i \Sigma y_i\right]\right)$$

correlation coefficient

$$r_{xy} = \frac{s_{xy}}{s_x s_y} = \frac{s_{xy}'}{\sigma x \sigma y}$$

NOTE: n is a positive integer and $n>1$.

EXAMPLE:

x_i	26	30	44	50	62	68	74
y_i	92	85	78	81	54	51	40

SOLUTION:

```
        GSB1
92.00   ENT↑
26.00   Σ+
85.00   ENT↑
30.00   Σ+
78.00   ENT↑
44.00   Σ+
81.00   ENT↑
50.00   Σ+
54.00   ENT↑
62.00   Σ+
51.00   ENT↑
68.00   Σ+
40.00   ENT↑
74.00   Σ+
        R/S
50.57   ***   x̄
        R/S
68.71   ***   ȳ
        R/S
18.50   ***   sx
        R/S
20.00   ***   sy
        R/S
-354.14 ***   sxy
        R/S
-0.96   ***   rxy
        GSB2
17.13   ***   σx
        R/S
18.51   ***   σy
        R/S
-303.55 ***   sxy'
        R/S
-0.96   ***   rxy
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Initialize		GSB	1	
3.	Perform steps 3-4 for i-1,2,...n	y_i	ENT↑		
4.		x_i	Σ+		
	(correct erroneous data)	y_k	ENT↑		
		x_k	f	Σ-	
5.	Calculate means:		R/S		$\bar{x}$
			R/S		$\bar{y}$
6a.	Calculate sample standard deviations:		R/S		S_x
			R/S		S_y
7a.	Calculate sample covarience		R/S		S_{xy}
8a.	Calculate correlation coefficient		R/S		r_{xy}
	AND OR:				
6b.	Calculate populations standard deviations:		GSB	2	σ_x
			R/S		σ_y
7b.	Calculate population covarience		R/S		S_{xy}
8b.	Calculate correlation coefficient		R/S		r_{xy}

Program Listings

Step	Key	Comment
01	*LBL1	Initialize
02	CLΣ	
03	R/S	
04	RC.0	
05	1	
06	-	n-1
07	STO1	
08	RC.0	
09	÷	
10	√X	
11	STO2	
12	$\bar{x}$	
13	R/S	*** $\bar{x}$
14	X⇄Y	
15	R/S	*** $\bar{y}$
16	S	
17	R/S	*** S_x
18	X⇄Y	
19	R/S	*** S_y
20	x	
21	*LBL0	
22	RC.5	
23	RC.1	
24	RC.3	
25	x	
26	RC.0	
27	÷	
28	-	
29	RCL1	n-1 or n
30	÷	
31	R/S	*** S_{xy} or S_{xy}'
32	X⇄Y	S_xS_y or $\sigma x \sigma y$
33	÷	
34	R/S	** r_{xy}
35	*LBL2	
36	S	
37	RCL2	
38	x	
39	R/S	***σ_x
40	X⇄Y	
41	RCL2	
42	x	
43	R/S	***σ_y
44	x	
45	RC.0	
46	STO1	$\sigma_x\sigma_y$
47	R↓	
48	GTO0	

** "Printx" may be inserted before "R/S".

*** "Printx" may be inserted before or in place of "R/S".

REGISTERS

0	1 n-1	2 $\sqrt{\frac{n-1}{n}}$	3	4	5
6	7	8	9	.0 n	.1 Σx
.2 Σx^2	.3 Σy	.4 Σy^2	.5 Σxy	16	17
18	19	20	21	22	23
24	25	26	27	28	29

ANALYSIS OF VARIANCE (ONE WAY)

The one-way analysis of variance tests the differences between the population means of k treatment groups. Group i (i= 1, 2, ..., k) has n_i observations (treatment group may have equal or unequal number of observations).

sum_i = sum of observations in treatment group i

$$= \sum_{j=1}^{n_i} x_{ij}$$

$$\text{Total SS} = \sum_{i=1}^{k} \sum_{j=1}^{n_i} X_{ij}^2 - \frac{\left(\sum_{i=1}^{k} \sum_{j=1}^{n_i} X_{ij}\right)^2}{\sum_{i=1}^{k} n_i}$$

$$\text{Treat SS} = \sum_{i=1}^{k} \left[\frac{\left(\sum_{j=1}^{n_i} X_{ij}\right)^2}{n_i} - \frac{\left(\sum_{i=1}^{k} \sum_{j=1}^{n_i} X_{ij}\right)^2}{\sum_{i=1}^{k} n_i} \right]$$

Error SS = Total SS - Treat SS

df_1 = Treat df = k-1

$$df_2 = \text{Error df} = \sum_{i=1}^{k} n_i - k$$

$$\text{Treat MS} = \frac{\text{Treat SS}}{\text{Treat df}}$$

$$\text{Error MS} = \frac{\text{Error SS}}{\text{Error df}}$$

$$F = \frac{\text{Treat MS}}{\text{Error MS}}$$ (with k - 1 and $\sum_{i=1}^{k} n_i$ - k degrees of freedom)

Total SS, Treat SS, Error SS are in registers R_1, R_2, R_3.

REFERENCE: Mathematical Statistics, J.E. Freund, Prentice Hall, 1962

EXAMPLE:

	i \ j	1	2	3	4	5	6
	1	10	8	5	12	14	11
Treatment	2	6	9	8	13		
	3	14	13	10	17	16	

SOLUTION:

```
        CLRG
10.00   Σ+
 8.00   Σ+
 5.00   Σ+
12.00   Σ+
14.00   Σ+
11.00   Σ+
        GSB1
60.00   ***   Sum1
 6.00   Σ+
 9.00   Σ+
 8.00   Σ+
13.00   Σ+
        GSB1
36.00   ***   Sum2
14.00   Σ+
13.00   Σ+
10.00   Σ+
17.00   Σ+
16.00   Σ+
        GSB1
70.00   ***   Sum3
        R/S
 2.00   ***   df1
        R/S
12.00   ***   df2
        R/S
 3.79   ***   F
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Initialize		f	REG	
3	Perform 3-5 for i= 1, 2, ...k				
4	Perform 4 for j= 1, 2, ...n_i	X_{ij}	Σ+		j
	(correct erroneous data, X_{im}	X_{im}	f	Σ-	
5	Calculate Sum		GSB	1	Sum_i
6	Calculate Treat df		R/S		df_1
7	Calculate Error df		R/S		df_2
8	Calculate F		R/S		F
9	(optional) Total SS		RCL	1	Total SS
10	(optional) Treat SS		RCL	2	Treat SS
11	(optional) Error SS		RCL	3	Error SS
12	For a new case, go to step 2				

Program Listings

```
01 *LBL1
02     1
03  ST+4
04  RC.1
05  ST+7
06    X²
07  RC.0
08  ST+5
09     ÷
10  ST+3
11     0
12  ST.0  }
13  RC.1  } restore registers
14  S-.1  }
15   R/S      **Sum_i
16 *LBL2
17  RC.2
18  RCL7
19    X²
20  RCL5
21     ÷
22     -
23  STO1      Total SS
24  RCL3
25  LSTX
26     -
27  STO2      Treat SS
28     -
29  STO3      Error SS
30  LSTX
31  RCL4
32     1
33     -
34   R/S      ***df1
35     ÷
36   X⇄Y
37  RCL5
38  RCL4
39     -
40   R/S      ***df2
41     ÷
42     ÷
43   R/S      ***F
```

** "Printx" may be inserted before "R/S".

***"Printx" may be inserted before or in place of "R/S".

REGISTERS

0	1 Total SS	2 Treat SS	3 Error SS	4 k	5 $\sum_{i=1}^{k} n_i$
6	7 $\sum_{i=1}^{k}\sum_{j=1}^{n_i} X_{ij}$	8	9	.0 n	.1 $\sum_{j=1}^{n_i} X_{ij}$
.2 $\sum_{i=1}^{k}\sum_{j=1}^{n_i} X_{ij}^2$		.4	.5	16	
		20	21	22	
	25	26	27	28	29

CHI-SQUARE DISTRIBUTION

This program evaluates the chi-square density

$$f(x)=\frac{1}{2^{\frac{\nu}{2}}\Gamma(\frac{\nu}{2})}x^{\frac{\nu}{2}-1}e^{-\frac{x}{2}}$$

where $x \geq 0$

ν is the degrees of freedom.



Series approximation is used to evaluate the cumulative distribution

$$P(x)=\int_0^x f(t)\,dt$$

$$=\left(\frac{x}{2}\right)^{\frac{\nu}{2}}\frac{e^{-\frac{x}{2}}}{\Gamma(\frac{\nu+2}{2})}\cdots$$

$$\cdots\left[1+\sum_{k=1}^{\infty}\frac{x^k}{(\nu+2)(\nu+4)\ldots(\nu+2k)}\right]$$

The program computes successive partial sums of the above series. When two consecutive partial sums are equal, the value is used as the sum of the series.

If ν is even,

$$\Gamma(\tfrac{\nu}{2}) = (\tfrac{\nu}{2} - 1)!$$

If ν is odd,

$$\Gamma(\tfrac{\nu}{2}) = (\tfrac{\nu}{2}-1)(\tfrac{\nu}{2}-2)\ldots(\tfrac{1}{2})\Gamma(\tfrac{1}{2})$$

$$\Gamma(\tfrac{1}{2}) = \sqrt{\pi}$$

REFERENCE: Handbook of Mathematical Functions. Abramowitz and Stegun, National Bureau of Standards, 1968

NOTES: 1. Program requires $\nu \leq 141$.

2. If both x and ν are large, f(x) may overflow the machine.

EXAMPLES:

1. ν= 20,
   x= 9.591;
   x= 15.

SOLUTION:

```
20.00 ENT↑
9.591 GSB1
 0.02  ***   f(x)
        R/S
 0.03  ***   P(x)
15.00 STO2
       GSB2
 0.06  ***   f(x)
        R/S
 0.22  ***   P(x)
```

EXAMPLE:

2. ν= 3,
   x= 7.82.

SOLUTION:

```
3.00 ENT↑
7.82 GSB1
0.02  ***  f(x)
      R/S
0.95  ***  P(x)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Enter degrees of freedom	ν	ENT↑		
3	Enter x	x			
4	Compute f(x)		GSB	1	f(x)
5	Compute P(x)		R/S		P(x)
6	For a different x:	x	STO	2	
7	Calculate f(x) and go to step 5		GSB	2	f(x)
8	For a new case, go to step 2				

Program Listings


```
*LBL1
STO2        x
R↓
1
STO3
X⇄Y         ν
2
÷
STO1
INT
LSTX
X≠Y?        ν even or odd?
GTO0        odd
1
X=Y?        Γ = 1
GTO2
-
STO3
*LBL5       factorial loop
1
X=Y?
GTO2
-
ST×3
GTO5
*LBL0
.
5
X=Y?
GTO9
X⇄Y
1
-
ST×3
GTO0
*LBL9
Pi
√X
ST×3
*LBL2
RCL2
RCL1
1
-
Yˣ
RCL2
2
÷
CHS
eˣ
×
2
RCL1
Yˣ
÷
RCL3
÷
STO5
R/S         ***f(x)
RCL2
RCL1
÷
ST×5
2
RCL1
×
STO6
1
STO4
*LBL8
RCL2
RCL6
2
+
STO6
÷
RCL4
×
STO4
+           add to 1 first time
X≠Y?
GTO8        loop
RCL5
×
R/S         ***P(x)
```

***"Printx" may replace "R/S".

REGISTERS

0	1 ν/2	2 x	3 1, Γ(ν/2)	4 used	5 used
6 used	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

t-DISTRIBUTION

This program evaluates the integral for t distribution

$$I(x,\nu) = \int_{-x}^{x} \frac{\Gamma(\frac{\nu+1}{2})\ (1+\frac{y^2}{\nu})^{-\frac{\nu+1}{2}}}{\sqrt{\pi\nu}\ \Gamma(\frac{\nu}{2})} dy$$

where $x>0$.

ν is the degrees of freedom.



EQUATIONS:

(1) ν even

$$I(x,\nu)=\sin\theta \left\{1+\frac{1}{2}\cos^2\theta+\frac{1\cdot 3}{2\cdot 4}\cos^4\theta+\ldots + \frac{1\cdot 3\cdot 5\ldots(\nu-3)}{2\cdot 4\cdot 6\ldots(\nu-2)}\cos^{\nu-2}\theta\right\}$$

(2) ν odd

$$I(x,\nu)=\begin{cases} \frac{2\theta}{\pi} & \text{if } \nu=1 \\ \frac{2\theta}{\pi}+\frac{2}{\pi}\cos\theta\left\{\sin\theta\left[1+\frac{2}{3}\cos^2\theta+\ldots + \frac{2\cdot 4\ldots(\nu-3)}{1\cdot 3\ldots(\nu-2)}\cos^{\nu-3}\theta\right]\right\} & \text{if } \nu>1 \end{cases}$$

where $\theta = \tan^{-1}(\frac{x}{\sqrt{\nu}})$

REFERENCE: Handbook of Mathematical Functions, Abramowitz and Stegun, National Bureau of Standards, 1968.

EXAMPLE:

I(2.201,11)

I(2.75,30)

SOLUTIONS:

```
2.201 ENT↑
11.00 GSB1
 0.95  ***   I(2.201,11)
 2.75 ENT↑
30.00 GSB1
 0.99  ***   I(2.75,30)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter x	x	ENT↑		
3.	Enter degrees of freedom	ν			
4.	Calculate $I(x,\nu)$		GSB	1	$I(x,\nu)$

Program Listings

Step	Comment	Step	Comment
01 *LBL1		48 ÷	
02 STO1		49 ST+6	
03 RAD		50 DSZ	
04 √X		51 GTO7	
05 ÷		52 RCL6	
06 TAN⁻¹		53 RCL4	
07 STO2	θ	54 x	
08 RCL1		55 RTN	
09 2		56 *LBL0	
10 ÷		57 RCL2	
11 INT		58 2	
12 LSTX		59 x	
13 X≠Y?	ν even or odd?	60 Pi	
14 GTO0	odd	61 ÷	
15 0		62 STO7	2θ/π
16 STO5		63 RCL1	
17 GSB9		64 1	
18 R/S	** I(x,ν)	65 STO5	
19 *LBL9		66 ST-1	
20 RCL2		67 X=Y?	ν = 1?
21 COS		68 GTO6	
22 X²		69 GSB9	
23 STO3	$\cos^2\theta$	70 RCL2	
24 RCL2		71 COS	
25 SIN		72 x	
26 STO4	sinθ	73 2	
27 RCL1		74 x	
28 2		75 Pi	
29 X=Y?	ν=2?	76 ÷	
30 GTO8		77 RCL7	
31 ÷		78 +	
32 1		79 R/S	** I(x,ν)
33 -		80 *LBL8	
34 STO0	i=INT $(\frac{\nu}{2} - 1)$	81 RCL4	
35 1		82 R/S	** I(x,ν)
36 STO6		83 *LBL6	
37 *LBL7		84 RCL7	
38 RCL3		85 R/S	** I(x,ν)
39 x			
40 RCL5			
41 1			
42 +			
43 x			
44 LSTX	R_5 + 1		
45 1			
46 +			
47 STO5			

** "Printx" may be inserted before "R/S".

REGISTERS

0 i	1 ν or ν-1	2 θ	3 $\cos^2\theta$	4 sinθ	5 Used
6 Used	7 2θ/π	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

F DISTRIBUTION

This program evaluates the integral of the F distribution

$$Q(x)=\int_x^{\infty}\frac{\Gamma\left(\frac{\nu_1+\nu_2}{2}\right)y^{\frac{\nu_1}{2}-1}\left(\frac{\nu_1}{\nu_2}\right)^{\frac{\nu_1}{2}}}{\Gamma\left(\frac{\nu_1}{2}\right)\Gamma\left(\frac{\nu_2}{2}\right)\left(1+\frac{\nu_1}{\nu_2}y\right)^{\frac{\nu_1+\nu_2}{2}}}dy$$

for given values of $x(>0)$, degrees of freedom ν_1,ν_2, provided either ν_1 or ν_2 is even.



The integral is evaluated by means of the following series:

(1) ν_1 even

$$Q(x)=t^{\frac{\nu_2}{2}}\left[1+\frac{\nu_2}{2}(1-t)+\dots+\frac{\nu_2(\nu_2+2)\dots(\nu_2+\nu_1-4)}{2\cdot4\dots(\nu_1-2)}(1-t)^{\frac{\nu_1-2}{2}}\right]$$

(2) ν_2 even

$$Q(x)=1-(1-t)^{\frac{\nu_1}{2}}\left[1+\frac{\nu_1}{2}t+\dots+\frac{\nu_1(\nu_1+2)\dots(\nu_2+\nu_1-4)}{2\cdot4\dots(\nu_2-2)}t^{\frac{\nu_2-2}{2}}\right]$$

where $t=\frac{\nu_2}{\nu_2+\nu_1 x}$

EXAMPLE:

1. $\nu_1=7$, $\nu_2=6$, $x=4.21$

SOLUTION:

```
4.21 ENT↑
7.00 ENT↑
6.00 GSB1
0.05  ***   Q(4.21)
```

EXAMPLE:

2. $\nu_1=4$, $\nu_2=20$, $x=2.25$

SOLUTION:

```
 2.25 ENT↑
 4.00 ENT↑
20.00 GSB1
 0.10  ***   Q(2.25)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Enter x	x	ENT↑		
3	Enter ν_1	ν_1	ENT↑		
4	Enter ν_2	ν_2			
5	Calculate Q (x)		GSB	1	Q(x)

Program Listings

Step	Key	Comment
01	*LBL1	
02	ENT↑	
03	R↓	
04	STO2	ν_2
05	R↓	
06	STO1	ν_1
07	x	
08	+	
09	÷	
10	STO3	t
11	RCL1	
12	2	
13	÷	
14	FRC	
15	X≠0?	ν_1 even or odd?
16	GTO9	ν_1 odd
17	GSB0	ν_1 even
18	R/S	**Q(x)
19	*LBL0	
20	RCL3	
21	RCL2	
22	2	
23	STO7	
24	÷	
25	Y^x	
26	STO4	
27	RCL1	
28	2	
29	-	
30	2	
31	÷	
32	STO0	$i = \frac{\nu_x - 2}{2}$
33	X=0?	
34	GTO8	
35	1	
36	STO5	
37	RCL3	
38	-	
39	STO3	1-t (or t)
40	RCL2	
41	2	
42	÷	
43	x	
44	ST+5	
45	DSZ	
46	GTO7	
47	GTO6	
48	*LBL7	
49	2	
50	ST+2	
51	ST÷7	
52	R↓	
53	RCL2	
54	RCL7	
55	÷	
56	RCL3	
57	x	
58	x	
59	ST+5	
60	DSZ	
61	GTO7	
62	*LBL6	
63	RCL5	
64	RCL4	
65	x	
66	RTN	
67	*LBL9	ν_2 even
68	RCL1	
69	RCL2	
70	STO1	
71	X⇄Y	
72	STO2	
73	1	
74	RCL3	
75	-	1-t
76	STO3	
77	GSB0	
78	1	
79	X⇄Y	
80	-	
81	R/S	**Q(x)
82	*LBL8	
83	RCL4	
84	R/S	**Q(x)

**"Printx" may be inserted before "R/S".

REGISTERS

0 i	1 ν_1 or ν_2	2 ν_2 or ν_1	3 t, 1-t	4 $t^{\frac{\nu_2}{2}}$ or $(1-t)^{\frac{\nu_1}{2}}$	5 used
6	7 used	8	9		.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

POISSON DISTRIBUTION

This program evaluates f(x) and P(x) for a given λ using the recursive relation

$$f(x+1) = \frac{\lambda}{x+1} f(x)$$

Density function

$$f(x) = \frac{\lambda^x e^{-\lambda}}{x!}$$

Cumulative distribution

$$P(x) = \sum_{k=0}^{x} f(k)$$

where

mean = variance = λ

λ is positive

and

x is a positive integer

EXAMPLE:

$\lambda = 3.2$, $x = 7$

SOLUTION:

```
3.20 GSB1
7.00  R/S
0.03  ***  f(7)
      R/S
0.98  ***  P(7)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Enter λ	λ	GSB	1	f(0)=P(0)= .04
3	Enter x	x			
4	Calculate f(x)		R/S		f(x)
5	Calculate P(x)		R/S		P(x)
6	For a new x, go to step 3				

Program Listings

01 *LBL1 02 STO1 03 CHS 04 e^x 05 STO2 06 R/S 07 STO5 08 0 09 STO0 10 RCL2 11 STO3 12 STO4 13 *LBL0 14 RCL1 15 ISZ 16 RCL0 17 ÷ 18 RCL3 19 x 20 STO3 21 ST+4 22 RCL0 23 RCL5 24 X≠Y? 25 GTO0 26 RCL3 27 R/S 28 1 29 RCL4 30 X>Y? 31 X⇄Y 32 R/S	 λ f(0) x ***f(x) ***P(x)		

***"Printx" may be inserted before or in place of "R/S".

REGISTERS

0 i	1 λ	2 f(0)	3 f(x)	4 P(x)	5 x
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

PARABOLIC CURVE FIT

For a set of data points $\{(x_i, y_i),\ i = 1,2,\ldots,n\}$ this program fits a parabola

$$y = a_2x^2 + a_1 x + a_0$$

with the sum of the squares of the errors minimized.

EQUATIONS: The normal equations are

$$\Sigma x^2y = a_2\Sigma x^4 + a_1\Sigma x^3 + a_0\Sigma x^2$$

$$\Sigma xy = a_2\Sigma x^3 + a_1\Sigma x^2 + a_0\Sigma x$$

$$\Sigma y = a_2\Sigma x^2 + a_1\Sigma x + a_0 n,$$

where the summations are from 1 to n.

NOTE: If $\Sigma x^3 = 0$, an error will occur. Replace it with 10^{-49}.

REFERENCE: "Applications Programs, Volume 1," Adams, Ed. Int'l. Software Clearinghouse, Estacada, Oregon 1976. pp. 15-18.

EXAMPLE:

x_i	0	1	1.5	3	5
y_i	2.1	2	-5	-24.5	-80

$\Sigma x_i^3 = 156.38$, $\Sigma x_i y_i = -479.00$,
$\Sigma x_i^2 y_i = -2229.75$

$\Sigma x_i^4 = 712.06$, $n = 5.00$, $\Sigma x_i^2 = 37.25$

$\Sigma x_i = 10.50$, $\Sigma y_i = -105.40$

SOLUTION:

```
         CLRG
   2.10  ENT↑
   0.00  GSB1
  20.00  ENT↑
   1.00  GSB1
  20.00  ENT↑   Correct erroneous
   1.00  GSB4        data
   2.00  ENT↑
   1.00  GSB1
  -5.00  ENT↑
   1.50  GSB1
 -24.50  ENT↑
   3.00  GSB1
 -80.00  ENT↑
   5.00  GSB1
         GSB2
  -3.66  ***    a2
         R/S
   1.85  ***    a1
         R/S
   2.28  ***    a0
   4.00  GSB3
 -48.83  ***    ŷ
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Initialize		f	REG	
3.	Perform 3-4 for i = 1,2,...n	y_i	ENT↑		
4.		x_i	GSB	1	i
	(correct erroneous data x_k, y_k)	y_k	ENT↑		
		x_k	GSB	4	i
5.	Calculate regression coefficients:		GSB	2	a_2
			R/S		a_1
			R/S		a_0
6.	Calculate estimated value, $\hat{y}$	x	GSB	3	$\hat{y}$
7.	For a new x, go to step 8				

Program Listings

```
01 *LBL1        x y
02    Σ+
03  LSTX
04  LSTX
05    3         x³
06    Yˣ
07  ST+8
08    x
09  ST+7        x⁴
10   √X
11    x
12  ST+9        x²y
13  RC.0        n
14   R/S
15 *LBL2        Solve 3 simultan-
16  RCL8          eous equations
17  RC.2
18  STO6
19  ST÷7
20  ST÷9
21    ÷
22  RC.1
23  S÷.2
24  ST÷8
25  S÷.5
26  RC.0
27  S÷.1
28  ST÷6
29  S÷.3
30   R↓
31   R↓
32  RC.1
33  S-.2
34    -
35    -
36  RCL6
37  ST-7
38  ST-8
39  RC.3
40  ST-9
41  S-.5
42  RC.2
43  ST÷8
44  S÷.5
45  LSTX
46  X=0?
47  GTO0
48  ST÷7
49  ST÷9
```

```
50  RCL8
51  ST-7
52  RC.5
53  ST-9
54 *LBL0
55  RCL7
56  ST÷9
57  RCL9
58  RCL8
59    x
60  S-.5
61  RCL9
62   R/S        *** a2
63  RCL6
64    x
65  S-.3
66  RC.5
67   R/S        *** a1
68  RC.1
69    x
70  S-.3
71  RC.3
72   R/S        *** a0
73 *LBL3        x
74  ENT↑
75   X²
76  RCL9
77    x
78   X⇄Y
79  RC.5
80    x
81    +
82  RC.3
83    +
84   R/S        ŷ manual "Printx" optional
85 *LBL4        Correction routine
86    Σ-
87  LSTX
88  LSTX
89    3
90    Yˣ
91  ST-8
92    x
93  ST-7
94   √X
95    x
96  ST-9
97  RC.0
98   R/S        n
```

*** "Printx" may replace "R/S".

REGISTERS

0	1	2	3	4	5
6 Used	7 Σx^4	8 Σx^3	9 $\Sigma x^2 y, a_2$	.0 n	.1 Σx
.2 Σx^2	.3 $\Sigma y, a_0$	.4 Used	.5 $\Sigma xy, a_1$	16	17
18	19	20	21	22	23
24	25	26	27	28	29

PAIRED t-STATISTIC

Given a set of paired observations from two normal populations with means μ_1, μ_2 (unknown)

x_i	x_1	x_2	...	x_n
y_i	y_1	y_2	...	y_n

let

$$D_i = x_i - y_i$$

$$\overline{D} = \frac{1}{n}\sum_{i=1}^{n} D_i$$

$$S_D = \sqrt{\frac{\Sigma D_i^2 - \frac{1}{n}(\Sigma D_i)^2}{n-1}}$$

$$S_{\overline{D}} = \frac{S_D}{\sqrt{n}}$$

The test statistic

$$t = \frac{\overline{D}}{S_{\overline{D}}}$$

which has n-1 degrees of freedom (df) c can be used to test the null hypothesis

$$H_0: \mu_1 = \mu_2$$

REFERENCE: Statistics in Research, B. Ostle, Iowa State University Press, 1963

EXAMPLE:

x_i	14	17.5	17	17.5	15.4
y_i	17	20.7	21.6	20.9	17.2

SOLUTION:

	CLΣ	
14.00	ENT↑	
17.00	GSB1	
17.50	ENT↑	
20.70	GSB1	
17.00	ENT↑	
21.60	GSB1	
17.50	ENT↑	
20.90	GSB1	
15.40	ENT↑	
17.20	GSB1	
	GSB2	
-3.20	***	$\overline{D}$
	R/S	
1.00	***	S_D
	R/S	
-7.16	***	t
	R/S	
4.00	***	df

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Initialize		f	E	
3.	Perform 3-4 for i - 1,2,...,n	x_i	ENT↑		
4.		y_i	GSB	1	i
	(correct erroneous data)	x_k	ENT↑		
		y_k	GSB	3	
5.	Compute statistics:		GSB	2	$\bar{D}$
			R/S		S_D
			R/S		t
			R/S		df

Program Listings

Program	Comments
01 *LBL1	y x
02 -	
03 Σ+	
04 R/S	
05 *LBL2	
06 $\bar{x}$	
07 R/S	*** $\overline{D}$
08 S	
09 LSTX	
10 X⇄Y	
11 R/S	*** S_D
12 RC.0	
13 √X	
14 ÷	
15 ÷	
16 R/S	*** t
17 RC.0	
18 1	
19 -	
20 R/S	*** df
21 *LBL3	
22 -	
23 Σ-	
24 R/S	Correct errors

*** "Printx" may be inserted before or in place of "R/S".

REGISTERS

0	1	2	3	4	5
6	7	8	9	.0 n	.1 Σx
.2 Σx^2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

t-STATISTIC FOR TWO MEANS

Suppose $x_1, x_2, \ldots x_{n_1}$ and $y_1, y_2, \ldots, y_{n_2}$ are independent random samples from two normal populations having means μ_1, μ_2 (unknown) and the same unknown variance σ^2.

We want to test the null hypothesis

$$H_0: \mu_1 - \mu_2 = D$$

Define

$$\bar{x} = \frac{1}{n_1} \sum_{i=1}^{n_1} x_i$$

$$\bar{y} = \frac{1}{n_2} \sum_{i=1}^{n_2} y_i$$

$$t = \frac{\bar{x} - \bar{y} - D}{\sqrt{\frac{1}{n_1} + \frac{1}{n_2}}} * \ldots$$

$$\frac{1}{\sqrt{\frac{\Sigma x_i^2 - n_1\bar{x}^2 + \Sigma y_i^2 - n_2\bar{y}^2}{n_1 + n_2 - 2}}}$$

We can use this t statistic which has the t distribution with $n_1 + n_2 - 2$ degrees of freedom (df) to test the null hypothesis H_0.

REFERENCE: Statistical Theory and Methodology in Science and Engineering, K.A. Brownlee, John Wiley & Sons, 1965.

EXAMPLE:

x: 79, 84, 108, 114, 120, 103, 122, 120

y: 91, 103, 90, 113, 108, 87, 100, 80, 99, 54

$n_1 = 8$

$n_2 = 10$

$D = 0$ (i.e., $H_0: \mu_1 = \mu_2$)

SOLUTION:

```
         CLRG
 79.00   Σ+
 84.00   Σ+
108.00   Σ+
114.00   Σ+
120.00   Σ+
103.00   Σ+
122.00   Σ+
120.00   Σ+
         GSB1
 91.00   Σ+
103.00   Σ+
 90.00   Σ+
113.00   Σ+
108.00   Σ+
 87.00   Σ+
100.00   Σ+
 80.00   Σ+
 99.00   Σ+
 54.00   Σ+
         GSB2
 16.00   ***   df
  0.00   R/S   D
  1.73   ***   t
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Initialize		f	REG	
3.	Perform 3 for i = 1, 2, ... n_1	x_i	Σ+		i
	(correct erroneous data)	x_k	f	Σ-	
4.	Reinitialize		GSB	1	
5.	Perform 5 for i - 1, 2, ... n_2	y_j	Σ+		i
	(correct erroneous data)	y_h	f	Σ-	
6.	Calculate statistics:				
6a.	Degrees of freedom		GSB	2	df
6b.	Enter D and calculate t statistic	D	R/S		t
7.	For a different D	D	GSB	3	t

Program Listings

Key	Comment
*LBL1	
RC.0	
STO4	
1/X	
STO1	
$\bar{x}$	
STO2	
S	
RC.0	
1	
-	
√X	
x	
STO3	
CLΣ	Reinitialize
R/S	
*LBL2	
RC.0	
ST+4	
1/X	
ST+1	
$\bar{x}$	
ST-2	
S	
RC.0	
1	
-	
√X	
x	
RCL3	
→P	$\sqrt{x^2 + y^2}$
RCL1	
RCL4	
2	
-	
R/S	** df
R↓	Save D
÷	
√X	
x	
STO5	Denominator
X⇄Y	D
*LBL3	
RCL2	
-	
CHS	
RCL5	
÷	
R/S	** t

** "Printx" may be inserted before "R/S".

REGISTERS

0	1 $1/n_1, 1/n_1+1/n_2$	2 $\bar{x}, \bar{x}-\bar{y}$	3 Used	4 n_1, n_1+n_2	5 Denominator
6	7	8	9	.0 n	.1 Σx
.2 Σx^2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

CHI-SQUARE EVALUATION

This program calculates the value of the x^2 statistic for the goodness of fit test by the equation

$$\chi^2 = \sum_{i=1}^{n} \frac{(O_i - E_i)^2}{E_i}$$

where O_i = observed frequency

E_i = expected frequency

If the expected values are equal

$$(E = E_i = \frac{\Sigma O_i}{n} \text{ for all } i)$$

then

$$\chi^2 = \frac{n \, \Sigma O_i^2}{\Sigma O_i} - \Sigma O_i$$

Note: In order to apply the goodness of fit test to a set of given data, combining some classes may be necessary to make sure that each expected frequency is not too small (say, not less than 5).

REFERENCES: Mathematical Statistics, J.E. Freund, Prentice Hall, 1962

EXAMPLE:

1.

O_i	8	50	47	56	5	14
E_i	9.6	46.75	51.85	54.4	8.25	9.15

SOLUTION:

```
         CLRG
  8.00 ENT↑
  9.60 GSB1
 50.00 ENT↑
 46.75 GSB1
 47.00 ENT↑
 51.85 GSB1
 56.00 ENT↑
 54.40 GSB1
  5.00 ENT↑
  8.25 GSB1
 14.00 ENT↑
  9.15 GSB1
  6.00 ENT↑
  9.00 GSB1   Correct erroneous data
  6.00 ENT↑
  9.00 GSB2
       GSB3
  4.84  ***   χ²
```

EXAMPLE:

2. The following table shows the observed frequencies in tossing a die 120 times. χ^2 can be used to test if the die is fair.

Note: Assume that the expected frequencies are equal.

number	1	2	3	4	5	6
frequency O_i	25	17	15	23	24	16

SOLUTION:

```
      CLRG
25.00 GSB1
17.00 GSB1
15.00 GSB1
23.00 GSB1
24.00 GSB1
16.00 GSB1
 9.00 GSB1   Correct erroneous data
 9.00 GSB2
      GSB4
 5.00  ***   χ2
       R/S
20.00  ***   E
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Initialize		f	REG	
3	For equal expected values:				
3a	Perform 3a for i= 1, 2, ...n	O_i	GSB	1	i
	(Correct erroneous data, O_j)	O_j	GSB	2	i
3b	Calculate χ^2		GSB	4	χ^2
3c	Calculate E		R/S		E
4	When expected values are unequal:				
4a	Perform 4a-4b for i= 1, 2, ...n	O_i	ENT ↑		
4b		E_i	GSB	1	i
	(Correct erroneous data, O_j, E_j)	O_j	ENT ↑		
		E_j	GSB	2	i
4c	Calculate χ^2		GSB	3	χ^2
5	For a new case, go to step 2				

Program Listings

01 *LBL1			
02 STO3	E_i or O_i		
03 -			
04 X^2			
05 RCL3			
06 ÷			
07 ST+2			
08 RCL3			
09 Σ+			
10 R/S	i		
11 *LBL2	correction routine		
12 STO3			
13 -			
14 X^2			
15 RCL3			
16 ÷			
17 ST-2			
18 RCL3			
19 Σ-			
20 R/S			
21 *LBL3			
22 RCL2			
23 R/S	** χ^2		
24 *LBL4			
25 RC.2	ΣO_i^2		
26 RC.0	n		
27 x			
28 RC.1	ΣO_i		
29 ÷			
30 LSTX			
31 -			
32 R/S	*** χ^2		
33 $\bar{x}$			
34 R/S	*** E		

** "Printx" may be inserted before "R/S".
***"Printx" may be inserted before or in place of "R/S".

REGISTERS

0	1	2 χ^2	3 used	4	5
6	7	8	9	.0 n	.1 Σx
.2 Σx^2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

2 x K CONTINGENCY TABLE

Contingency tables can be used to test the null hypothesis that two variables are independent.

	1	2	3	...	k	Totals
A	a_1	a_2	a_3	...	a_k	N_A
B	b_1	b_2	b_3	...	b_k	N_B
Totals	N_1	N_2	N_3	...	N_k	N

Test statistic

$$\chi^2 = \frac{N}{N_A} \sum_{i=1}^{k} \frac{a_i^2}{N_i} + \frac{N}{N_B} \sum_{i=1}^{k} \frac{b_i^2}{N_i} - N$$

Degrees of freedom df = k - 1

Pearson's coefficient of contingency C measures the degree of association between the two variables

$$C = \sqrt{\frac{\chi^2}{N+\chi^2}}$$

REFERENCE: Statistics in Research, B. Ostle, Iowa State University Press, 1963

EXAMPLE:

	1	2	3
A	2	5	4
B	3	8	7

SOLUTION:

```
     CLRG
2.00 ENT↑
3.00 GSB1
5.00 ENT↑
8.00 GSB1
4.00 ENT↑
7.00 GSB1
9.00 ENT↑
8.00 GSB1
9.00 ENT↑
8.00 GSB2   correct error
     R/S
2.00 ***    df
     R/S
0.02 ***    χ²
     R/S
0.03 ***    C
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1	Key in the program				
2	Initialize		f	REG	
3	Perform 3-4 for i= 1, 2, ...k	a_i	ENT↑		
4		b_i	GSB	1	i
	(correct erroneous data a_j, b_j)	a_j	ENT↑		
		b_j	GSB	2	i
5	Calculate degreesof freedom		R/S		df
6	Calculate χ^2		R/S		χ^2
7	Calculate Pearson's coefficient of		R/S		C
	contingency				

Program Listings

01	*LBL1	b a
02	Σ+	
03	LSTX	b a
04	*LBL0	
05	+	N_i
06	+	same in last x
07	RC.2	b^2
08	LSTX	
09	÷	
10	ST+1	
11	RC.4	a^2
12	LSTX	
13	÷	
14	ST+2	
15	0	
16	ST.2	
17	ST.4	
18	RC.0	i
19	R/S	
20	1	
21	-	
22	R/S	***df
23	RC.1	N_B
24	RC.3	N_A
25	+	N
26	ENT↑	
27	ST×1	
28	ST×2	
29	CHS	
30	RCL1	
31	RC.1	
32	÷	
33	+	
34	RCL2	
35	RC.3	
36	÷	
37	+	
38	R/S	***χ^2
39	+	
40	LSTX	
41	X⇄Y	
42	÷	
43	√X	
44	R/S	***C
45	*LBL2	correction routine
46	Σ-	
47	LSTX	b a
48	GTO0	

***"Printx" may be inserted before or in place of "R/S".

REGISTERS

0	1 $\Sigma \frac{b_i^2}{N_i}$	2 $\Sigma \frac{a_i^2}{N_i}$	3	4	5
6			9	.0 i	.1 N_B
.2 b^2	.3 N_A	.4 a^2	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

In the Hewlett-Packard tradition of supporting HP programmable calculators with quality software, the following titles have been carefully selected to offer useful solutions to many of the most often encountered problems in your field of interest. These ready-made programs are provided with convenient instructions that will allow flexibility of use and efficient operation. We hope that these Solutions books will save your valuable time. They provide you with a tool that will multiply the power of your HP-19C or HP-29C many times over in the months or years ahead.

Mathematics Solutions
Statistics Solutions
Financial Solutions
Electrical Engineering Solutions
Surveying Solutions
Games
Navigational Solutions
Civil Engineering Solutions
Mechanical Engineering Solutions
Student Engineering Solutions